Problem Chapter I.7

The planar R-RTR-RTR mechanism considered is shown in Fig. 1. The driver link is the rigid link 1 (the link AB). The following numerical data are given: $AB = 0.140$ m, $AC = 0.060$ m, $AE = 0.250$ m, $CD = 0.150$ m. The angle of the driver link 1 with the horizontal axis is $\phi = 30°$. The constant angular speed of the driver link 1 is 50 rpm.

Position analysis for an input angle

Position of joint A

A Cartesian reference frame xOy is selected. The joint A is in the origin of the reference frame, that is, $A \equiv O$,

$$x_A = 0, \;\; y_A = 0. \tag{1}$$

Position of joint C

The coordinates of the joint C are

$$x_C = 0, \;\; y_C = AC = 0.060 \text{ m}. \tag{2}$$

Position of joint E

The coordinates of the joint E are

$$x_E = 0, \;\; y_E = -AE = -0.250 \text{ m}. \tag{3}$$

Position of joint B

The unknowns are the coordinates of the joint B, x_B and y_B. Because the joint A is fixed and the angle ϕ is known, the coordinates of the joint B are computed from the following expressions

$$\begin{aligned} x_B &= AB \cos\phi = 0.140 \cos 30° = 0.121 \text{ m}, \\ y_B &= AB \sin\phi = 0.140 \sin 30° = 0.070 \text{ m}. \end{aligned} \tag{4}$$

Position of joint D

The unknowns are the coordinates of the joint D, x_D and y_D.
The joint D is located on the line BC:

$$\begin{aligned} &\frac{y_D - y_C}{x_D - x_C} = \frac{y_B - y_C}{x_B - x_C} \quad \text{or} \\ &(x_B - x_C)(y_D - y_C) = (x_D - x_C)(y_B - y_C) \quad \text{or} \\ &(0.121 - 0)(y_D - 0.060) = (x_D - 0)(0.070 - 0.060). \end{aligned} \tag{5}$$

Furthermore, the length of the segment CD is constant

$$\begin{aligned}(x_C - x_D)^2 + (y_C - y_D)^2 = CD^2 \quad \text{or} \\ (0 - x_D)^2 + (0.060 - y_D)^2 = 0.150^2. \end{aligned} \tag{6}$$

The Eqs. (5) and (6) form a system from which the coordinates of the joint D can be computed. Two sets of solutions are found for the position of the joint D. These solutions are located at the intersection of the line BC with the circle centered in C and radius CD (Fig. 2), and they have the following numerical values:

$$x_{D1} = -0.149 \text{ m}, \; y_{D1} = 0.047 \text{ m},$$
$$x_{D2} = 0.149 \text{ m}, \; y_{D2} = 0.072 \text{ m}.$$

To determine the correct position of the joint D for the mechanism, an additional condition is needed.

For the first quadrant, $0 \le \phi \le 90°$, the condition is $x_D \le x_C$.

Because $x_C = 0$ m, the coordinates of the joint D are

$$x_D = x_{D1} = -0.149 \, \text{m},$$
$$y_D = y_{D1} = 0.047 \, \text{m}.$$

Angle ϕ_2

The angle of link 2 (or link 3) with the horizontal axis is calculated from the slope of the straight line BC:

$$\phi_2 = \phi_3 = \arctan \frac{y_B - y_C}{x_B - x_C} = \arctan \frac{0.070 - 0.060}{0.121 - 0} = 0.082 \text{ rad} = 4.715°.$$

Angle ϕ_4

The angle of link 5 (or link 4) with the horizontal axis is obtained from the slope of the straight line ED:

$$\phi_4 = \phi_5 = \arctan \frac{y_E - y_D}{x_E - x_D} = \arctan \frac{-0.250 - 0.047}{0 + 0.149} = -1.105 \, \text{rad} = -63.333°.$$

Velocity and Acceleration Analysis

Algebraic Method

The velocity of the point $B = B_1$ on the link 1 is

$$\mathbf{v}_B = \mathbf{v}_{B_1} = \mathbf{v}_A + \boldsymbol{\omega}_1 \times \mathbf{r}_{AB} = \boldsymbol{\omega}_1 \times \mathbf{r}_B,$$

where $\mathbf{v}_A \equiv \mathbf{0}$ is the velocity of the origin $A \equiv O$.
The angular velocity of link 1 is

$$\boldsymbol{\omega}_1 = \omega_1 \mathbf{k} = \frac{\pi n}{30}\mathbf{k} = \frac{\pi(50)}{30}\mathbf{k} = 5.235\mathbf{k} \ \text{ rad/s.}$$

the position vector of point B is

$$\mathbf{r}_{AB} = \mathbf{r}_B - \mathbf{r}_A = \mathbf{r}_B = x_B\boldsymbol{\imath} + y_B\boldsymbol{\jmath} + z_B\mathbf{k} = 0.121\boldsymbol{\imath} + 0.070\boldsymbol{\jmath} \ \text{ m.}$$

The velocity of point B_2 on the link 2 is $\mathbf{v}_{B_2} = \mathbf{v}_{B_1}$ because between the links 1 and 2 there is a rotational joint.

The velocity of $B_1 = B_2$ is

$$\mathbf{v}_{B_1} = \mathbf{v}_{B_2} = \begin{vmatrix} \boldsymbol{\imath} & \boldsymbol{\jmath} & \mathbf{k} \\ 0 & 0 & \omega \\ x_B & y_B & 0 \end{vmatrix} = \begin{vmatrix} \boldsymbol{\imath} & \boldsymbol{\jmath} & \mathbf{k} \\ 0 & 0 & 5.235 \\ 0.121 & 0.070 & 0 \end{vmatrix} = -0.366\boldsymbol{\imath} + 0.634\boldsymbol{\jmath} \ \text{ m/s.}$$

The acceleration of the point $B = B_1$ on the link 1 is

$$\begin{aligned} \mathbf{a}_B &= \mathbf{a}_{B_1} = \mathbf{a}_{B_2} = \mathbf{a}_A + \boldsymbol{\alpha}_1 \times \mathbf{r}_B + \boldsymbol{\omega}_1 \times (\boldsymbol{\omega}_1 \times \mathbf{r}_B) = \boldsymbol{\alpha}_1 \times \mathbf{r}_B - \boldsymbol{\omega}_1^2 \mathbf{r}_B \\ &= -\boldsymbol{\omega}_1^2 \mathbf{r}_B = -5.235^2(0.121\boldsymbol{\imath} + 0.070\boldsymbol{\jmath}) = -3.323\boldsymbol{\imath} - 1.919\boldsymbol{\jmath} \ \text{ m/s}^2. \end{aligned}$$

The angular acceleration of link 1 is $\boldsymbol{\alpha}_1 = \dot{\boldsymbol{\omega}}_1 = \mathbf{0}$.

The velocity of the point B_3 on the link 3 is calculated in terms of the velocity of the point B_2 on the link 2

$$\mathbf{v}_{B_3} = \mathbf{v}_{B_2} + \mathbf{v}^{rel}_{B_3B_2} = \mathbf{v}_{B_2} + \mathbf{v}_{B_{32}}, \tag{7}$$

where $\mathbf{v}^{rel}_{B_3B_2} = \mathbf{v}_{B_{32}}$ is the relative acceleration of B_3 with respect to B_2 on link 3. This relative velocity is parallel to the sliding direction BC, $\mathbf{v}_{B_{32}} || BC$, or

$$\mathbf{v}_{B_{32}} = v_{B_{32}} \cos\phi_2 \boldsymbol{\imath} + v_{B_{32}} \sin\phi_2 \boldsymbol{\jmath}, \tag{8}$$

where $\phi_2 = 4.715°$ is known from position analysis. The points B_3 and C are on the link 3 and

$$\mathbf{v}_{B_3} = \mathbf{v}_C + \boldsymbol{\omega}_3 \times \mathbf{r}_{CB} = \boldsymbol{\omega}_3 \times (\mathbf{r}_B - \mathbf{r}_C), \tag{9}$$

where $\mathbf{v}_C \equiv \mathbf{0}$ and the angular velocity of link 3 is

$$\boldsymbol{\omega}_3 = \omega_3 \mathbf{k}.$$

Equations (7), (8), and (9) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ x_B - x_C & y_B - y_C & 0 \end{vmatrix} = \mathbf{v}_{B_2} + v_{B_{32}} \cos\phi_2 \mathbf{\imath} + v_{B_{32}} \sin\phi_2 \mathbf{\jmath}. \tag{10}$$

Equation (10) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns ω_3 and $v_{B_{32}}$

$$-\omega_3(y_B - y_C) = v_{Bx} + v_{B_{32}} \cos\phi_2,$$
$$\omega_3(x_B - x_C) = v_{By} + v_{B_{32}} \sin\phi_2,$$

or

$$-\omega_3(0.070 - 0.060) = -0.366 + v_{B_{32}} \cos 4.715°,$$
$$\omega_3(0.121 - 0) = 0.634 + v_{B_{32}} \sin 4.715°.$$

It results

$$\omega_3 = \omega_2 = 5.448 \text{ rad/s} \quad \text{and} \quad v_{B_{32}} = 0.313 \text{ m/s}.$$

The acceleration of the point B_3 on the link 3 is calculated in terms of the acceleration of the point B_2 on the link 2

$$\mathbf{a}_{B_3} = \mathbf{a}_{B_2} + \mathbf{a}^{rel}_{B_3B_2} + \mathbf{a}^{cor}_{B_3B_2} = \mathbf{a}_{B_2} + \mathbf{a}_{B_{32}} + \mathbf{a}^{cor}_{B_{32}}, \tag{11}$$

where $\mathbf{a}^{rel}_{B_3B_2} = \mathbf{a}_{B_{32}}$ is the relative acceleration of B_3 with respect to B_2 on link 3. This relative acceleration is parallel to the sliding direction BC, $\mathbf{a}_{B_{32}} || BC$, or

$$\mathbf{a}_{B_{32}} = a_{B_{32}} \cos\phi_2 \mathbf{\imath} + a_{B_{32}} \sin\phi_2 \mathbf{\jmath}. \tag{12}$$

The Coriolis acceleration of B_3 realative to B_2 is

$$\mathbf{a}^{cor}_{B_{32}} = 2\ \boldsymbol{\omega}_3 \times \mathbf{v}_{B_{32}} = 2\ \boldsymbol{\omega}_2 \times \mathbf{v}_{B_{32}} = 2 \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ v_{B_{32}} \cos\phi_2 & v_{B_{32}} \sin\phi_2 & 0 \end{vmatrix} =$$
$$2(-\omega_3 v_{B_{32}} \sin\phi_2 \mathbf{\imath} + \omega_3 v_{B_{32}} \cos\phi_2 \mathbf{\jmath}) =$$
$$2[-5.448(0.313)\sin 4.715^\circ \mathbf{\imath} + 5.448(0.313)\cos 4.715^\circ \mathbf{\jmath}] =$$
$$-0.280\mathbf{\imath} + 3.400\mathbf{\jmath} \ \ \text{m/s}^2. \tag{13}$$

The points B_3 and C are on the link 3 and

$$\mathbf{a}_{B_3} = \mathbf{a}_C + \boldsymbol{\alpha}_3 \times \mathbf{r}_{CB} - \omega_3^2 \mathbf{r}_{CB}, \tag{14}$$

where $\mathbf{a}_C \equiv \mathbf{0}$ and the angular acceleration of link 3 is

$$\boldsymbol{\alpha}_3 = \alpha_3 \mathbf{k}.$$

Equations (11), (12), (13), and (14) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \alpha_3 \\ x_B - x_C & y_B - y_C & 0 \end{vmatrix} - \omega_3^2(\mathbf{r}_B - \mathbf{r}_C) =$$
$$\mathbf{a}_{B_2} + a_{B_{32}}(\cos\phi_2 \mathbf{\imath} + \sin\phi_2 \mathbf{\jmath}) + 2\ \boldsymbol{\omega}_3 \times \mathbf{v}_{B_{32}}. \tag{15}$$

Equation (15) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns α_3 and $a_{B_{32}}$

$$-\alpha_3(y_B - y_C) - \omega_3^2(x_B - x_C) = a_{Bx} + a_{B_{32}} \cos\phi_2 - 2\omega_3 v_{B_{32}} \sin\phi_2,$$
$$\alpha_3(x_B - x_C) - \omega_3^2(y_B - y_C) = a_{By} + a_{B_{32}} \sin\phi_2 + 2\omega_3 v_{B_{32}} \cos\phi_2,$$

or

$$-\alpha_3(0.070 - 0.060) - 5.448^2(0.121 - 0) =$$
$$-3.323 + a_{B_{32}} \cos 4.715^\circ - 2(5.448)(0.313)\sin 4.715^\circ,$$
$$\alpha_3(0.121 - 0) - 5.448^2(0.070 - 0.060) =$$
$$-1.919 + a_{B_{32}} \sin 4.715^\circ + 2(5.448)(0.313)\cos 4.715^\circ.$$

It results

$$\alpha_3 = \alpha_2 = 14.568 \text{ rad/s}^2 \ \ \text{and} \ \ a_{B_{32}} = -0.140 \ \ \text{m/s}^2.$$

The velocity of D_3 is

$$\mathbf{v}_{D_3} = \mathbf{v}_{D_4} = \mathbf{v}_C + \boldsymbol{\omega}_3 \times \mathbf{r}_{CD} = \boldsymbol{\omega}_3 \times (\mathbf{r}_D - \mathbf{r}_C) =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ x_D - x_C & y_D - y_C & 0 \end{vmatrix} = \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & 5.448 \\ -0.149 - 0 & 0.047 - 0.060 & 0 \end{vmatrix} =$$
$$0.067\mathbf{\imath} - 0.814\mathbf{\jmath} \text{ m/s}.$$

The acceleration of D_3 is

$$\mathbf{a}_{D_3} = \mathbf{a}_{D_4} = \mathbf{a}_C + \boldsymbol{\alpha}_3 \times \mathbf{r}_{CD} - \omega_3^2 \mathbf{r}_{CD} = \boldsymbol{\alpha}_3 \times (\mathbf{r}_D - \mathbf{r}_C) - \omega_3^2(\mathbf{r}_D - \mathbf{r}_C) =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \alpha_3 \\ x_D - x_C & y_D - y_C & 0 \end{vmatrix} - \omega_3^2 \left[(x_D - x_C)\mathbf{\imath} + (y_D - y_C)\mathbf{\jmath}\right] =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & 14.568 \\ -0.149 - 0 & 0.047 - 0.060 & 0 \end{vmatrix} -$$
$$5.448^2 \left[(-0.149 - 0)\mathbf{\imath} + (0.047 - 0.060)\mathbf{\jmath}\right] =$$
$$4.617\mathbf{\imath} - 1.811\mathbf{\jmath} \text{ m/s}^2.$$

The velocity of the point D_5 on the link 5 is calculated in terms of the velocity of the point D_4 on the link 4

$$\mathbf{v}_{D_5} = \mathbf{v}_{D_4} + \mathbf{v}_{D_{54}}. \tag{16}$$

This relative velocity of D_5 with respect to D_4 is parallel to the sliding direction DE, $\mathbf{v}_{D_{54}} || DE$, or

$$\mathbf{v}_{D_{54}} = v_{D_{54}} \cos\phi_5 \mathbf{\imath} + v_{D_{54}} \sin\phi_5 \mathbf{\jmath}. \tag{17}$$

The points D_5 and E are on the link 5 and

$$\mathbf{v}_{D_5} = \mathbf{v}_E + \boldsymbol{\omega}_5 \times \mathbf{r}_{ED} = \boldsymbol{\omega}_5 \times (\mathbf{r}_D - \mathbf{r}_E), \tag{18}$$

where $\mathbf{v}_E \equiv \mathbf{0}$ and the angular velocity of link 5 is

$$\boldsymbol{\omega}_5 = \omega_5 \mathbf{k}.$$

Equations (16), (17), and (18) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_5 \\ x_D - x_E & y_D - y_E & 0 \end{vmatrix} = \mathbf{v}_{D_4} + v_{D_{54}}(\cos\phi_5 \mathbf{\imath} + \sin\phi_5 \mathbf{\jmath}). \tag{19}$$

Equation (19) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns ω_5 and $v_{D_{54}}$

$$-\omega_5(y_D - y_E) = v_{D4x} + v_{D_{54}} \cos\phi_5,$$
$$\omega_5(x_D - x_E) = v_{D4y} + v_{D_{54}} \sin\phi_5,$$

or

$$-\omega_5(0.047 - 0.250) = 0.067 + v_{D_{54}} \cos(-63.333°),$$
$$\omega_5(-0.149 - 0) = -0.814 + v_{D_{54}} \sin(-63.333°).$$

It results

$$\omega_5 = \omega_4 = 0.917 \text{ rad/s} \ \text{ and } \ v_{D_{54}} = -0.757 \ \text{ m/s}.$$

The acceleration of the point D_5 on the link 5 is calculated in terms of the acceleration of the point D_4 on the link 4

$$\mathbf{a}_{D_5} = \mathbf{a}_{D_4} + \mathbf{a}_{D_{54}} + \mathbf{a}^{cor}_{D_{54}}, \tag{20}$$

This relative acceleration $\mathbf{a}_{B_{32}}$ is parallel to the sliding direction DE, $\mathbf{a}_{D_{54}}||DE$, or

$$\mathbf{a}_{D_{54}} = a_{D_{54}} \cos\phi_5 \mathbf{\imath} + a_{D_{54}} \sin\phi_5 \mathbf{\jmath}. \tag{21}$$

The Coriolis acceleration of D_5 realative to D_4 is

$$\mathbf{a}^{cor}_{D_{54}} = 2\ \boldsymbol{\omega}_4 \times \mathbf{v}_{D_{54}} = 2\ \boldsymbol{\omega}_5 \times \mathbf{v}_{D_{54}} = 2 \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_5 \\ v_{D_{54}} \cos\phi_5 & v_{D_{54}} \sin\phi_5 & 0 \end{vmatrix} =$$
$$2(-\omega_5 v_{D_{54}} \sin\phi_5 \mathbf{\imath} + \omega_5 v_{D_{54}} \cos\phi_5 \mathbf{\jmath}) =$$
$$2[-0.917(-0.757)\sin(-63.333°)\mathbf{\imath} + 0.917(-0.757)\cos(-63.333°)\mathbf{\jmath}] =$$
$$-0.280\mathbf{\imath} + 3.400\mathbf{\jmath} \ \text{ m/s}^2. \tag{22}$$

The points D_5 and E are on the link 5 and

$$\mathbf{a}_{D_5} = \mathbf{a}_E + \boldsymbol{\alpha}_5 \times \mathbf{r}_{ED} - \omega_5^2 \mathbf{r}_{ED}, \tag{23}$$

where $\mathbf{a}_E \equiv \mathbf{0}$ and the angular acceleration of link 5 is

$$\boldsymbol{\alpha}_5 = \alpha_5 \mathbf{k}.$$

Equations (20), (21), (22), and (23) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \alpha_5 \\ x_D - x_E & y_D - y_E & 0 \end{vmatrix} - \omega_5^2(\mathbf{r}_D - \mathbf{r}_E) =$$
$$\mathbf{a}_{D_4} + a_{D_{54}}(\cos\phi_5\mathbf{\imath} + \sin\phi_5\mathbf{\jmath}) + 2\ \boldsymbol{\omega}_5 \times \mathbf{v}_{D_{54}}. \tag{24}$$

Equation (24) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns α_5 and $a_{D_{54}}$

$$-\alpha_5(y_D - y_E) - \omega_5^2(x_D - x_E) = a_{D4x} + a_{D_{54}}\cos\phi_5 - 2\omega_5 v_{D_{54}}\sin\phi_5,$$
$$\alpha_5(x_D - x_E) - 2\omega_5^2(y_D - y_E) = a_{D4y} + a_{D_{54}}\sin\phi_5 + 2\omega_5 v_{D_{54}}\cos\phi_5,$$

or

$$-\alpha_5(0.047 - 0.250) - 0.917^2(-0.149 - 0) =$$
$$4.617 + a_{D_{54}}\cos(-63.333°) - 2(0.917)(-0.757)\sin(-63.333°),$$
$$\alpha_5(-0.149 - 0) - 0.917^2(0.047 - 0.250) =$$
$$-1.811 + a_{D_{54}}\sin(-63.333°) + 2(0.917)(-0.757)\cos(-63.333°).$$

It results

$$\alpha_5 = \alpha_4 = -5.771 \text{ rad/s}^2 \quad \text{and} \quad a_{D_{54}} = 3.411 \text{ m/s}^2.$$

The *Mathematica*^{TM} program for the velocity and acceleration analysis using the algebraic method is given in Program 4-II.



Fig. 1



Fig. 2

```
(* VELOCITY AND ACCELERATION ANALYSIS *)
(* Algebraic Method *)

Apply [Clear, Names["Global`*"] ] ;
Off[General::spell];
Off[General::spell1];

AB=0.14;
AC=0.06;
AE=0.25;
CD=0.15;
ϕ=π/6.;

xA = yA = 0;
rA = { xA, yA, 0} ;

xC = 0 ;
yC = AC ;
rC = { xC, yC, 0} ;

xE = 0 ;
yE = -AE ;
rE = { xE, yE, 0} ;

xB = AB Cos[ϕ];
yB = AB Sin[ϕ];
rB = { xB, yB, 0} ;

eqnD1 = ( xDs - xC )^2 + ( yDs - yC )^2 - CD^2 == 0 ;
eqnD2 = ( yB - yC )( xDs - xC ) == ( yDs - yC )( xB - xC ) ;
solutionD = Solve [ { eqnD1 , eqnD2 } , { xDs, yDs } ];
xD1 = xDs /. solutionD[[1]];
yD1 = yDs /. solutionD[[1]];
xD2 = xDs /. solutionD[[2]];
yD2 = yDs /. solutionD[[2]];
If [ xD1 <= xC , xD = xD1 ; yD = yD1 , xD = xD2 ; yD=yD2 ] ;
rD = { xD, yD, 0} ;

ϕ2=ϕ3=ArcTan[(yB-yC)/(xB-xC)];
ϕ4=ϕ5=ArcTan[(yD-yE)/(xD-xE)];

markers = Table [ {
            Point [ { xA , yA } ] ,
            Point [ { xB , yB } ] ,
            Point [ { xC , yC } ] ,
            Point [ { xD , yD } ] ,
            Point [ { xE , yE } ]
            } ] ;

name = Table [ {
            Text [ "A" ,{0  , 0  },{-1 , 1 } ] ,
            Text [ "B" ,{xB , yB },{ 0 ,-1 } ] ,
            Text [ "C" ,{xC , yC },{ -1,-1 } ] ,
            Text [ "D" ,{xD , yD },{ 0 ,-1 } ] ,
            Text [ "E" ,{xE , yE },{ -1, 1 } ]
            } ] ;

graph = Graphics [
        { { RGBColor [ 1 , 0 , 0 ] ,
            Line [ { {xA,yA},{xB,yB} } ] } ,
          { RGBColor [ 0 , 1 , 0 ] ,
            Line [ { {xB,yB} , {xD,yD}} ] } ,
          { RGBColor [ 0 , 0 , 1 ] ,
```

```
            Line [ { {xD,yD},  {xE,yE}} ] } ,
          { RGBColor [ 1 , 1 , 1 ] ,
            PointSize [ 0.01 ] , markers } ,
          { name } } ] ;

Show [ Graphics [ graph ] ,
       PlotRange -> { { -.25 , .25 } ,
                      { -.3 , .25 } } ,
       Frame -> True,
       AxesOrigin -> {xA,yA},
       FrameLabel -> {"x","y"},
       Axes -> {True,True},
       AspectRatio -> Automatic ] ;


n = 50.; (* rpm *)
ω = ω1 = { 0, 0, n π/30 } ;
α = α1 = D[ω,t] ;
Print["ω = ω1 = ",ω," rad/s"];
Print["α = α1 = ",α," rad/s^2"];

vB=vB1=vB2=Cross[ω1,rB];
Print["vB = vB1 = ω1 x rB = ",vB," m/s"];
Print["vB1 = vB2 "];


aB=aB1=aB2=Cross[α1,rB]-ω1.ω1 rB;
Print["aB = aB1 = α1 x rB - ω1^2 rB = ",aB," m/s^2"];
Print["aB1 = aB2 "];

(*--ω3--*)

ω3u={ 0, 0, ω3z};
Print["ω3 = ω2 = ",ω3u];

vB3u=Cross[ω3u,rB-rC];
Print["vB3 = vC + ω3 x (rB-rC)"];

vB32u={ v32 Cos[ϕ2], v32 Sin[ϕ2], 0};
Print["vB32={vB32 Cos[ϕ2],vB32 Sin[ϕ2],0}"];

Print["vB3 = vB2 + vB32 => ω3z, vB32"];

eqvB=vB3u-vB2-vB32u;
solutionvB=Solve[{eqvB[[1]]==0,eqvB[[2]]==0},{ω3z,v32}];
ω3=ω2=ω3u/.solutionvB[[1]];
vB32=vB32u/.solutionvB[[1]];
Print["ω3 = ω2 = ",ω3," rad/s"];
Print["vB32 = ",v32/.solutionvB[[1]]," m/s"];
Print["vB32v = ",vB32," m/s"];

vB3=vB3u/.solutionvB[[1]];
Print["vB3 = ",vB3," m/s"];

vD3=vD4=Cross[ω3,rD-rC];
Print["vD3 = vD4 = vC + ω3 x (rD-rC) = ",vD3," m/s"];

(*--α3--*)

α3u={ 0, 0, α3z};
Print["α3 = α2 = ",α3u];

aB3u=Cross[α3u,rB-rC]-ω3.ω3(rB-rC) ;
Print["aB3 = aC + α3 x (rB-rC) - ω3.ω3(rB-rC)"];
```

```
aB32u={ a32 Cos[ϕ2], a32 Sin[ϕ2], 0};
Print["aB32={aB32 Cos[ϕ2],aB32 Sin[ϕ2],0}"];

aB32cor = 2 Cross[ω3,vB32];
Print["aB32cor = 2 ω3 x vB32 = ",aB32cor," m/s^2"];

Print["aB3 = aB2 + aB32 + aB32cor => α3z, aB32"];

eqaB=aB3u-aB2-aB32u-aB32cor;
solutionaB=Solve[{eqaB[[1]]==0,eqaB[[2]]==0},{α3z,a32}];
α3=α2=α3u/.solutionaB[[1]];
aB32=aB32u/.solutionaB[[1]];
Print["α3 = α2 = ",α3," rad/s^2"];
Print["aB32 = ",a32/.solutionaB[[1]]," m/s^2"];
Print["aB32v = ",aB32," m/s^2"];

aB3=aB3u/.solutionaB[[1]];
Print["aB3 = ",aB3," m/s^2"];

aD3=aD4=Cross[α3,rD-rC]-ω3.ω3(rD-rC) ;
Print["aD3 = aD4 = aC + α3 x (rD-rC) - ω3.ω3(rD-rC)"];
Print["aD3 = ",aD3," m/s^2"];

(*--ω5--*)

ω5u={ 0, 0, ω5z};
Print["ω5 = ω5 = ",ω5u];

vD5u=Cross[ω5u,rD-rE];
Print["vD5 = vE + ω5 x (rD-rE)"];

vD54u={ v54 Cos[ϕ4], v54 Sin[ϕ4], 0};
Print["vD54={vD54 Cos[ϕ4],vD54 Sin[ϕ4],0}"];

Print["vD5 = vD4 + vD54 => ω5z, vD54"];

eqvD=vD5u-vD4-vD54u;
solutionvD=Solve[{eqvD[[1]]==0,eqvD[[2]]==0},{ω5z,v54}];
ω5=ω4=ω5u/.solutionvD[[1]];
vD54=vD54u/.solutionvD[[1]];
Print["ω5 = ω4 = ",ω5," rad/s"];
Print["vD54 = ",v54/.solutionvD[[1]]," m/s"];
Print["vD54v = ",vD54," m/s"];

vD5=vD5u/.solutionvD[[1]];
Print["vD5 = ",vD5," m/s"];

(*--α5--*)

α5u={ 0, 0, α5z};
Print["α5 = α4 = ",α5u];

aD5u=Cross[α5u,rD-rE]-ω5.ω5(rD-rE) ;
Print["aD5 = aE + α5 x (rD-rE) - ω5.ω5(rD-rE)"];

aD54u={ a54 Cos[ϕ5], a54 Sin[ϕ5], 0};
Print["aD54={aD54 Cos[ϕ5],aD54 Sin[ϕ5],0}"];

aD54cor = 2 Cross[ω5,vD54];
Print["aD54cor = 2 ω5 x vD54 = ",aD54cor," m/s^2"];

Print["aD5 = aD4 + aD54 + aD54cor => α5z, aD54"];

eqaD=aD5u-aD4-aD54u-aD54cor;
solutionaD=Solve[{eqaD[[1]]==0,eqaD[[2]]==0},{α5z,a54}];
```

```
α5=α4=α5u/.solutionaD[[1]];
aD54=aD54u/.solutionaD[[1]];
Print["α5 = α4 = ",α5," rad/s^2"];
Print["aD54 = ",a54/.solutionaD[[1]]," m/s^2"];
Print["aD54v = ",aD54," m/s^2"];

aD5=aD5u/.solutionaD[[1]];
Print["aD5 = ",aD5," m/s^2"];
```




```
ω = ω1 = {0, 0, 5.23599} rad/s

α = α1 = {0, 0, 0} rad/s^2

vB = vB1 = ω1 x rB = {-0.366519, 0.63483, 0.} m/s

vB1 = vB2

aB = aB1 = α1 x rB - ω1^2 rB = {-3.32396, -1.91909, 0.} m/s^2

aB1 = aB2

ω3 = ω2 = {0, 0, ω3z}

vB3 = vC + ω3 x (rB-rC)

vB32={vB32 Cos[φ2],vB32 Sin[φ2],0}

vB3 = vB2 + vB32 => ω3z, vB32

ω3 = ω2 = {0, 0, 5.44826} rad/s

vB32 = 0.313096 m/s

vB32v = {0.312037, 0.0257363, 0} m/s

vB3 = {-0.0544826, 0.660566, 0.} m/s

vD3 = vD4 = vC + ω3 x (rD-rC) = {0.0671766, -0.814473, 0.} m/s

α3 = α2 = {0, 0, α3z}
```

```
aB3 = aC + α3 x (rB-rC) - ω3.ω3(rB-rC)

aB32={aB32 Cos[ϕ2],aB32 Sin[ϕ2],0}

aB32cor = 2 ω3 x vB32 = {-0.280436, 3.40011, 0.} m/s^2

aB3 = aB2 + aB32 + aB32cor => α3z, aB32

α3 = α2 = {0, 0, 14.5681} rad/s^2

aB32 = -0.140694 m/s^2

aB32v = {-0.140218, -0.011565, 0} m/s^2

aB3 = {-3.74462, 1.46946, 0.} m/s^2

aD3 = aD4 = aC + α3 x (rD-rC) - ω3.ω3(rD-rC)

aD3 = {4.61708, -1.81183, 0.} m/s^2

ω5 = ω5 = {0, 0, ω5z}

vD5 = vE + ω5 x (rD-rE)

vD54={vD54 Cos[ϕ4],vD54 Sin[ϕ4],0}

vD5 = vD4 + vD54 => ω5z, vD54

ω5 = ω4 = {0, 0, 0.917134} rad/s

vD54 = -0.757991 m/s

vD54v = {-0.34018, 0.677368, 0} m/s

vD5 = {-0.273003, -0.137105, 0.} m/s

α5 = α4 = {0, 0, α5z}

aD5 = aE + α5 x (rD-rE) - ω5.ω5(rD-rE)

aD54={aD54 Cos[ϕ5],aD54 Sin[ϕ5],0}

aD54cor = 2 ω5 x vD54 = {-1.24248, -0.623982, 0.} m/s^2

aD5 = aD4 + aD54 + aD54cor => α5z, aD54

α5 = α4 = {0, 0, -5.77155} rad/s^2

aD54 = -3.41104 m/s^2

aD54v = {-1.53085, 3.04823, 0} m/s^2

aD5 = {1.84376, 0.612421, 0.} m/s^2
```